

RAYLEIGH SCATTERING BY DIELECTRIC BODIES*

Purpose: To compute the tensor elements specifying the low frequency (Rayleigh) scattering of a plane electromagnetic wave by a homogeneous dielectric body with axial symmetry.

Language: Fortran IV

Authors: Thomas B.A. Senior and Thomas M. Willis III, Department of Electrical and Computer Engineering, University of Michigan, Ann Arbor, MI 48109

Availability: ASIS-NAPS document NAPS-

Description: When a plane electromagnetic wave is incident on a homogeneous dielectric body immersed in free space, the leading (Rayleigh) term in the low frequency expansion of the far zone scattered field can be expressed in terms of the electric and magnetic polarizability tensors $\bar{\bar{P}}$ and $\bar{\bar{M}}$. These are functions of the relative permittivity ϵ_r and relative permeability μ_r , respectively, and of the geometry of the body, but are independent of the excitation. If the body is symmetric about the x_3 axis of a Cartesian coordinate system x_1, x_2, x_3 , the tensors diagonalize and $P_{22} = P_{11}$, $M_{22} = M_{11}$. As shown in [1], $\bar{\bar{P}}$ and $\bar{\bar{M}}$ are particular cases of a general polarizability tensor $\bar{\bar{X}}(\tau)$ with $\bar{\bar{P}} = \bar{\bar{X}}(\epsilon_r)$, $\bar{\bar{M}} = -\bar{\bar{X}}(\mu_r)$. The tensor elements are weighted integrals of potential functions which are obtained by solution of the corresponding integral equation by the moment method. The program computes $X_{11}(\tau)$ and $X_{33}(\tau)$, corresponding to the transverse and axial components of the dipole moments respectively, for any complex τ ,

* This work was supported by the U.S. Army Armament Research and Development Command, Chemical Systems Laboratory, under Contract DAAK11-81-K-0004.

and is a modified version of an analogous program [2.3] for perfect conductivity ($\epsilon_r = \infty$, $\mu_r = 0$).

Although it is difficult to state the highest frequency at which the Rayleigh scattered field is applicable, a reasonable criterion is that the free space and interior wavelengths should both exceed the maximum linear dimensions of the body by a factor of 10 or more.

The body is described by its profile in the $z(=x_3), \rho$ plane of a cylindrical coordinate system. The profile is assumed to be composed of up to 15 straight line or circular arc segments. The former are described by their end points in the z, ρ plane, while the latter are described by their end points, the angles subtended at their centers of curvature and their concavities. Each segment is uniformly subdivided into cells and the total number N of cells across the entire profile must not exceed 102. For many shapes however, a much smaller value of N is adequate. Concave (re-entrant) shapes can be treated, as well as multiply connected ones, e.g., a torus. Bodies having two or more disjoint portions can also be considered provided their material parameters are the same and they have a common axis of symmetry.

An input data set consists of one control card and a number of segment specification cards, one for each segment of the profile. The control card lists:

SEGMENTS: number (≤ 15) of segments
EXCLUSION: a decimal fraction involved in the computation of a
 self cell; defaults to 0.001
TAU: complex material parameter
PRINT: prints (PRINT = 1) or suppresses (PRINT = 0) the two
 potentials obtained by solution of the integral
 equation; defaults to 0

A segment specification card lists:

CELLS: number of sampling cells on the segment

TYPE KEY: type of segment: 1 circular arc, concave up; 2 circular arc, concave down; 3 straight line

VOLUME SENSE: + on blank: additive volume;
 -: subtractive volume

Z-COORD.

ENDPOINTS: self-explanatory

RHO-COORD.

ENDPOINTS: self-explanatory

THETA(DEG.): angle subtended by a circular arc at its center of curvature

As an illustration, consider a sphere of radius 0.5 and $\tau = 2+i$ with $N = 20$. The corresponding profile is a semicircle in the z, ρ plane of radius 0.5 centered (say) at (0.5,0). The print out is as follows:

```
>***** PROGRAM DIELCOM *****  
  
>        SEGMENTS        = 1  
>        EXCLUSION      = 0.0010  
>        TAU             = 0.200000E+01 +J0.100000E+01  
  
>SEGMENT # 1#  
>        CELLS           = 20  
>        TYPE KEY        = 1  
>        VOLUME SENSE    = +  
>        1ST END POINT = (    0.0        ,    0.0        )  
>        2ND END POINT = (    1.00000,    0.0        )  
>        THETA (DEG)    = 180.00000  
  
>COMPUTED RESULTS#  
>        VOLUME        =    0.52360  
>        X11/V        =    0.88155 +J    0.52841  
>        X33/V        =    0.88297 +J    0.52928  
> #
```

The known (exact) results are $X_{11}/V = X_{33}/V = 0.88235\dots + i0.52941\dots$. In the sample run the imaginary part of X_{11}/V was computed with least accuracy (0.2 percent). The c.p.u. time for the Amdahl 470/V8 computer at the University of Michigan was 0.14 s. Increasing N to 30 reduced the errors by about a factor 2, but at the expense of a factor 2 increase in the c.p.u. time.

References

- [1] T.B.A. Senior, "Low frequency scattering by a dielectric body," Radio Sci., Vol. 11, pp. 477-482, 1976.
- [2] T.B.A. Senior and D. J. Ahlgren, "Rayleigh scattering," IEEE Antennas Propag. (Comp. Prog. Descr.), Vol. 21, p. 134, 1973.
- [3] T.B.A. Senior and D. J. Ahlgren, "The numerical solution of low frequency scattering problems," The University of Michigan Radiation Laboratory Report No. 031630-9-T, February 1972.

COMPUTER PROGRAM: DIELCOM

T.B.A. Senior and T. M. Willis III

Radiation Laboratory
 Department of Electrical and Computer Engineering
 The University of Michigan
 Ann Arbor, Michigan 48109

1. Introduction

The program solves two integral equations and then computes the two independent elements X_{11} and X_{33} of the general polarizability tensor $\bar{\bar{X}}$ for a homogeneous dielectric body whose material parameter τ , representing either the relative permittivity or relative permeability, may be complex. The body is assumed symmetric about the x_3 ($=z$) axis of a Cartesian coordinate system x_1, x_2, x_3 . The theoretical background and the definition of the tensor $\bar{\bar{X}}$ are given in [1], and the program itself is a modified version of one previously developed [2] for a perfectly conducting body. The numerical procedures will be discussed and the various subroutines presented along with a flow chart of the program.

2. Formulation2.1 Body Description

It is convenient to introduce a cylindrical polar coordinate system ρ, ϕ, z whose z axis is the axis of symmetry of the body. The profile of the body in the ρ, z plane is defined as $\rho = \rho(z)$, and is constructed of segments which may be linear arcs or circular arcs of either concavity.

Distance along the profile is measured by the variable s , and the tangent at any point forms the angle α with the z axis: $-\pi < \alpha \leq \pi$.

The volume of rotation of a linear arc (Type 3) segment whose endpoints in the z, ρ plane are (z_1, ρ_1) , (z_2, ρ_2) is

$$\begin{aligned} \delta V &= \int_{z_1}^{z_2} \pi \rho^2(z) dz \\ &= \frac{\pi}{2} (z_2 - z_1) (\rho_2^2 + \rho_1 \rho_2 + \rho_1^2) . \end{aligned} \quad (1)$$

For a circular arc which subtends an angle θ at its center (z_0, ρ_0) and is of radius a , let

$$\xi = \begin{cases} -1 & \text{if segment is concave down (Type 1)} \\ 1 & \text{if segment is concave up (Type 2)} \end{cases}$$

Then

$$\begin{aligned} \delta V &= (z_2 - z_1) \left\{ \rho_0^2 + a^2 - \frac{1}{3} (u_1^2 + u_1 u_2 + u_2^2) \right\} \\ &+ \rho_0 \left\{ u_2 (\rho_2 - \rho_0) - u_1 (\rho_1 - \rho_0) - \xi da^2 \theta \right\} , \end{aligned} \quad (2)$$

where

$$d = \frac{z_2 - z_1}{|z_2 - z_1|}$$

and

$$u_2 = z_2 - z_0 , \quad u_1 = z_1 - z_0 .$$

2.2 Integral Equations

Given the profile $\rho = \rho(z)$ and the volume V , quantities m and m_1 are defined as

$$m = m(z, \rho, z', \rho') = \frac{4\rho\rho'}{(\rho - \rho')^2 + (z - z')^2} \quad (3)$$

and

$$m_1 = 1 - m, \quad (4)$$

where (z, ρ) and (z', ρ') are distinct points on the profile. The unprimed variables denote the observation point and the primed variables the point of integration. Note that $0 \leq m, m_1 \leq 1$.

If $K(m)$ and $E(m)$ are the complete elliptic integrals of the first and second kinds respectively, let

$$\Omega_0 = \frac{1}{4} \left(\frac{m}{\rho\rho'} \right)^{3/2} \left\{ K(m) + 2mK'(m) \right\} \quad (5)$$

$$\Omega_1 = \left(\frac{m}{\rho\rho'} \right)^{3/2} \left\{ \left(1 - \frac{m}{2} \right) K'(m) - \frac{1}{4} K(m) \right\} \quad (6)$$

$$\Omega_2 = \frac{1}{m^2} \left(\frac{m}{\rho\rho'} \right)^{3/2} \left\{ 2m \left(1 - \frac{m}{2} \right)^2 K'(m) - \left(1 - \frac{m^2}{4} \right) K(m) + E(m) \right\} \quad (7)$$

where $K'(m) = (d/dm)K(m)$. For a body whose complex material parameter is τ , the integral equations to be solved are (see [1,3])

$$\frac{1 + \tau}{1 - \tau} W_1(s) = 2\rho + \frac{1}{\pi} \int \overline{W_1(s')} \left\{ \rho \cos \alpha' \Omega_2 + [(z' - z) \sin \alpha' - \rho' \cos \alpha'] \Omega_1 \right\} \rho' ds' \quad (8)$$

and

$$\frac{1 + \tau}{1 - \tau} W_3(s) = 2z + \frac{1}{\pi} \int \overline{W_3(s')} \left\{ \rho \cos \alpha' \Omega_1 + [(z' - z) \sin \alpha' - \rho' \cos \alpha'] \Omega_0 \right\} \rho' ds' \quad , \quad (9)$$

where the integration is over the entire profile of the body and the bar across the integral sign implies a Cauchy principal value. In terms of the solutions,

$$\frac{X_{11}}{V} = - \frac{\pi}{V} \int \rho' W_1(s') \cos \alpha' ds' \quad (10)$$

$$\frac{X_{33}}{V} = - \frac{2\pi}{V} \int \rho' W_3(s') \sin \alpha' ds' \quad (11)$$

where V is the volume of the body.

3. Numerical Solution of the Integral Equations

The numerical procedures involved in computing X_{33}/V are similar to those required for X_{11}/V , and it is therefore sufficient to give the details only for the latter.

3.1 X_{11}/V Computation

The primary task is the solution of the integral equation (8) for the function $W_1(s)$ and this entails the determination of a sequence of values $W_1^{(i)}$, $i = 1, 2, \dots, N$, approximating $W_1(s)$ at the sampling points $s = s_i$ on the profile $\rho = \rho(z)$. For this purpose the profile is divided into N cells C_i of arc length Δs_i and midpoints s_i corresponding to the coordinates (z_i, ρ_i) . Within each cell we also define the points s_{i-} and s_{i+} where

$$\left. \begin{aligned} s_{i-} &= s_i - \alpha_0 \Delta s_i \\ s_{i+} &= s_i + \alpha_0 \Delta s_i \end{aligned} \right\} \quad (12)$$

with the restriction $0 \leq \alpha_0 \leq 1/2$.

By assuming that $W_1(s)$ has the constant value $W_1^{(j)}$ over the j th cell, the integral on the right-hand side of (8) can be evaluated as a linear combination of the $W_1^{(j)}$ whose coefficients depend on the position (z, ρ) of the field point, leading to a linear system of N equations in N unknowns, viz.

$$\begin{aligned} - \sum_{j=1}^N W_1^{(j)} \int_{C_j} \rho_i \cos \alpha' \Omega_2 + [(z' - z_i) \sin \alpha' - \rho' \cos \alpha'] \Omega_1 \rho' ds' \\ = 2\pi \rho_i - \pi \frac{1 + \tau}{1 - \tau} W_1^{(i)} \end{aligned} \quad (13)$$

for $i = 1, 2, \dots, N$. Hence, the system to be solved is

$$Aw_1 = b \quad (14)$$

where w_1 is a column vector with elements

$$w_{1j} = W_1^{(j)}, \quad j = 1, 2, \dots, N,$$

A is a square matrix with elements

$$a_{ij} = - \int_{C_j} f(s') \rho' ds' \quad (i \neq j) \quad (15)$$

$$a_{ij} = - \int_{C_j} f(s') \rho' ds' + \pi \frac{1 + \tau}{1 - \tau} \quad (i = j) \quad (16)$$

where

$$f(s') = \rho_i \cos \alpha' \Omega_2 + [(z' - z_i) \sin \alpha' - \rho' \cos \alpha'] \Omega_1$$

and b is a row vector with elements

$$b_j = 2\pi\rho_j, \quad j = 1, 2, \dots, N.$$

The integrations were performed using three point Gaussian quadrature for which the parameter α_0 in (12) is $\alpha_0 = 1/2 \sqrt{3/5}$.

For $i \neq j$, we then have

$$a_{ij} = - \left[\frac{5}{18} \left\{ \rho_{j-} f(i, j-) + \rho_{j+} f(i, j+) \right\} + \frac{4}{9} f(i, j) \right] \Delta s_j \quad (17)$$

for $i, j = 1, 2, \dots, N$, in which

$$f(i,j) = \rho_i \cos \alpha_j \Omega_2(i,j) + \left\{ (z_j - z_i) \sin \alpha_j - \rho_j \cos \alpha_j \right\} \Omega_1(i,j) , \quad (18)$$

where the subscripts j^- and j^+ correspond to the points s_{j^-} and s_{j^+} of (12). The diagonal elements a_{ij} of the matrix A are

$$a_{ii} = \pi \frac{1 + \tau}{1 - \tau} - \int_{s_i - (1/2)\Delta s_i}^{s_i + (1/2)\Delta s_i} \rho' f(s') ds'$$

and require the numerical determination of a Cauchy principal value. As an approximation to this principal value, we remove from the cell C_i a slice defined by the interval $(s_i - (1/2)\beta\Delta s_i, s_i + (1/2)\beta\Delta s_i)$ where β , $0 < \beta \leq 1$, is the fractional exclusion; $\beta = 0$ implies no exclusion, i.e., that the principal value is not taken.

We now have

$$a_{ii} \approx \pi \frac{1 + \tau}{1 - \tau} - \int_{s_i - (1/2)\Delta s_i}^{s_i - (\beta/2)\Delta s_i} \rho' f(s') ds' - \int_{s_i + (\beta/2)\Delta s_i}^{s_i + (1/2)\Delta s_i} \rho' f(s') ds' \quad (19)$$

and these integrals are also computed using three-point Gaussian quadrature. Defining

$$\left. \begin{aligned}
 s_{i2} &= s_i - \frac{1}{4} (1 + \beta) \Delta s_i \\
 s_{i1} &= s_{i2} - \frac{1}{2} \alpha_o (1 - \beta) \Delta s_i \\
 s_{i3} &= s_{i2} + \frac{1}{2} \alpha_o (1 - \beta) \Delta s_i \\
 s_{i5} &= s_i + \frac{1}{4} (1 + \beta) \Delta s_i \\
 s_{i4} &= s_{i5} - \frac{1}{2} \alpha_o (1 - \beta) \Delta s_i \\
 s_{i6} &= s_{i5} + \frac{1}{2} \alpha_o (1 - \beta) \Delta s_i
 \end{aligned} \right\} \quad (20)$$

we obtain

$$\begin{aligned}
 a_{ii} &= \pi \frac{1 + \tau}{1 - \tau} - \frac{1}{2} (1 - \beta) \left[\frac{5}{18} \left\{ \rho_{i1} f(i, i_1) + \rho_{i3} f(i, i_3) + \rho_{i4} f(i, i_4) \right. \right. \\
 &\quad \left. \left. + \rho_{i6} f(i, i_6) \right\} + \frac{4}{9} \left\{ \rho_{i2} f(i, i_2) + \rho_{i5} f(i, i_5) \right\} \right] \Delta s_i \quad . \quad (21)
 \end{aligned}$$

Equations (17) through (21) completely describe a system of N linear equations in N unknowns $W_1^{(i)}$, $i = 1, 2, \dots, N$. Having determined the sampled values $W_1^{(i)} = W_1(s_i)$, X_{11}/V is computed from (10) by integration over each segment of the profile using a second order integration procedure (subroutine INTEG).

3.2 X_{33}/V Computation

The procedure for solving (8) and then computing X_{33}/V from (11) is about identical to that described above. The matrix equation for the sampled values $W_3(s_i) = W_3^{(i)}$ is $Aw_3 = b$ where

$$\begin{aligned} w_{3j} &= W_3^{(j)} \\ b_j &= 2\pi z_j \end{aligned}, \quad j = 1, 2, \dots, N, \quad (22)$$

and for $i, j = 1, 2, \dots, N$ with $i \neq j$,

$$a_{ij} = - \left[\frac{5}{18} \left\{ \rho_{j-} g(i, j-) + \rho_{j+} g(i, j+) \right\} + \frac{4}{9} g(i, j) \right] \Delta s_i \quad (23)$$

with

$$g(i, j) = \rho_i \cos \alpha_j \Omega_1(i, j) + \left\{ (z_j - z_i) \sin \alpha_j - \rho_j \cos \alpha_j \right\} \Omega_0(i, j). \quad (24)$$

The same approximation as before is used to compute the diagonal elements a_{ii} of the matrix A, and thus

$$\begin{aligned} a_{ii} &= \pi \frac{1 + \tau}{1 - \tau} - \frac{1}{2} (1 - \beta) \left[\frac{5}{18} \left\{ \rho_{i_1} g(i, i_1) + \rho_{i_3} g(i, i_3) + \rho_{i_4} g(i, i_4) \right. \right. \\ &\quad \left. \left. + \rho_{i_6} g(i, i_6) \right\} + \frac{4}{9} \left\{ \rho_{i_2} g(i, i_2) + \rho_{i_5} g(i, i_5) \right\} \right] \Delta s_i \end{aligned} \quad (25)$$

for $i = 1, 2, \dots, N$.

Equations (22) through (25) completely describe a system of N linear equations in N unknowns $W_3^{(j)}$, $j = 1, 2, \dots, N$. Once the sampled values $W_3^{(j)} = W_3(s_j)$ are found, X_{33}/V is computed from (11) by integration over each segment of the profile using a second order integration routine as before.

4. Computer Program

The program computes X_{11}/V and X_{33}/V , and consists of a main program and six subroutines.

4.1 Data Set

A data set is made up of a control card and a number of segment specification cards, one for each segment (or sub-segment) of the profile.

Control Card

<u>Columns</u>	<u>Description</u>
1-5	Integer number of segments in the body (must be <15).
6-17	The fraction exclusion β . If these columns are blank, β defaults to 0.001.
18-41	Complex material parameter τ .
42-46	Integer print key; 1 : print W_1 and W_3 potential functions 0 : do not print W_1 and W_3 (default is 0).

Segment Specification Card

<u>Columns</u>	<u>Description</u>
1-5	Integer (right justified): the number of sampling points or cells on the segment.
6-10	Segment type key: (right justified) 1 : circular arc, concave down 2 : circular arc, concave up 3 : linear.
11-13	Volume sense: (single character, right justified) + or blank: additive volume - : subtractive volume.
14-25, 26-37	Two real numbers: respectively, the end coordinates z_1 and z_2 of the segment.
38-49, 50-61	Two real numbers: respectively, the end coordinates ρ_1 and ρ_2 of the segment.
62-73	A real number: for circular arcs, the included angle in degrees.

There are the following restrictions:

- (i) the total number of segments must not exceed 15 and
- (ii) the total number of cells over all segments must not exceed 102.

The profile is specified in the direction of increasing profile-length, beginning either at its left-hand intersection with the z-axis and ending at its right-hand intersection with the z-axis, or beginning and ending at its own left-most point. Re-entrant segments are permitted, allowing


$z_1 > z_2$, and the "body" may consist of two or more disjoint parts.

For maximum accuracy in the X_{33} computation the body should be approximately centered about the origin.

4.2 Main Program

The main program reads, prints and supervises all computations.

A rough flow chart showing the interaction of the subroutines is given below.



4.3 Subroutine DATA (IN, MX, MY, ZEP, RHOEP, THETA, B, VOL)

This subroutine is called once for each segment of the profile. From the input specification for the segment, DATA computes the (z, ρ) coordinates of the necessary sampling points on the profile, the quantities $\cos \alpha$ and $\sin \alpha$ at these points and the incremental volume of the segment.

Comments:

Stored in COMMON are the arrays RHO(102,9), Z(102,9), ARC(102), C(102,9) and S(102,9) which contain the numbers computed by DATA.

For the Ith cell, the subscripts (I,J) correspond to the points s_{ij} of (41) when $1 \leq J \leq 6$. For $J = 7, 8, 9$, the subscripts (I,J) refer to the points s_{i-} , s_i and s_{i+} respectively of (12).

4.4 Subroutine INTEG (V, NSEG, NUMPTS, SUM)

INTEG numerically integrates quadratic interpolating polynomials approximating the complex valued data on each segment of the profile. When the profile is composed of several segments, no interpolation is performed across segment boundaries. Hence, the integration is accurate even for disconnected segments, e.g., the circular arcs of two spheres.

Comments:

Stored in COMMON are the lengths ARC (I), $I = 1, \dots, N$ required to compute the integral.

4.5 Subroutines CDLUD (N,ADIM,A,IV) and CDBS (N,TDIM,T,IV,B,X)

Used together, CDLUD and CDBS solve the linear system $AX = B$. CDLUD performs a L-U decomposition of the $N \times N$ matrix A and CDBS performs back-substitution. These routines are adapted from Forsythe and Moler [3].

4.6 Subroutine ELLI (M1, K, E, KPR)

This computes the elliptic integrals $K(m)$ and $E(m)$ and the derivative $K'(m)$ from their power series approximations [4].

4.7 Subroutine SETUP (I, J, L, AXI11, AXI33)

This is essential in computing the linear systems. Specifically SETUP, after calling ELLI, computes the quantities Ω_0 (Eq. 5), Ω_1 (Eq. 6), and Ω_2 (Eq. 7). These are then used to compute $f(i,j)$ (AXI11) of Eq. (18) and $g(i,j)$ (AXI33) of Eq. (24).

A listing of the entire program follows:

```

C*****C
C*****C
C
C   DIELCOM                               DEC. 1981
C

```

```

C*****C
C*****C
C
C   The PROGRAM DIELCOM computes the elements of the
C   polarizability tensor for a rotationally symmetric
C   homogeneous dielectric body. See "Radio Science", vol. 11
C   No. 5, pages 477-482, May 1976: Low-frequency scattering
C   by a dielectric body, by T.B.A. Senior
C
C   For a discussion of a similar program (MINK) see The
C   University of Michigan's Radiation Laboratory report,
C   "The Numerical Solution of Low Frequency Scattering
C   Problems", by Thomas B. A. Senior and David J. Ahlgren;
C   report no. 013630-9-T.
C

```

```

C*****C
C
C   LOGICAL DEVICE ASSIGNMENT
C
C   5:  FORMATTED INPUT PARAMETERS (SEE INPUT FORMAT)
C   6:  FORMATTED OUTPUT OF TENSOR ELEMENTS
C

```

```

C*****C
C
C   INPUT FORMAT
C
C   CARD#1  FORMAT(I5,3E12.6,I5)
C           NSEGS      NUMBER OF SEGMENTS IN THE BODY
C                   (NSEGS MUST BE <=15)
C           FR         FRACTIONAL EXCLUSION (BETA) USED WITH CELLS
C           TAU        COMPLEX MATERIAL PARAMETER
C           PRINTW     INTEGER PRINT KEY; PRINTW=1 PRINTS THE W
C                   POTENTIALS, PRINTW=0 DOES NOT (DEFAULT IS 0)
C
C   CARD#2 TO CARD#1+NS1+NS2  FORMAT(2I5,2X,A1,5E12.6)
C           NUMPTS     NUMBER OF CELLS THIS SEGMENT IS TO BE DIVIDED
C                   UP INTO (THE SUM OF ALL NUMPTS MUST BE <=102)
C           ITYP       TYPE OF SEGMENT; ITYP=1 CONCAVE DOWN SEGMENT,
C                   ITYP=2 CONCAVE UP SEGMENT, ITYP=3 LINEAR
C           ISIGN      CHARACTER CORRESPONDING TO THE VOLUME SENSE OF
C                   THIS SEGMENT; ISIGN='+' POSITIVE VOLUME,
C                   ISIGN='-' NEGATIVE VOLUME (DEFAULT IS '+')
C           ZEP(1)     Z CO-ORDINATE OF THE 1ST END POINT OF SEGMENT
C           ZEP(2)     Z CO-ORDINATE OF THE 2ND END POINT OF SEGMENT
C           RHOEP(1)   RHO CO-ORDINATE OF THE 1ST END POINT
C           RHOEP(2)   RHO CO-ORDINATE OF THE 2ND END POINT
C           THETA      ANGLE (IN DEGREES) SUBTENDED BY SEGMENT
C

```

```

C*****C
C   COMMON AX11(102,102),AX33(102,102),X(102),TAU,POM,X11,X33
C   DIMENSION B(102),ZEP(2),RHOEP(2),ST1(6),ST3(6)
C   INTEGER NUMPTS(15),PLUS/'+'/ ,BL/' '/ ,INDX(2),PRINTW
C   COMMON RHO(102,9),Z(102,9),ARC(102),C(102,9),S(102,9)
C   COMMON /SOL/ IPS(102)

```

DATA MIN,TWOPI,PI,W0,W1/'-',6.283185,3.141593,.4444444,.2777777/

C
CCC READ, CHECK, AND WRITE INPUT VALUES

C
37 READ(5,34,END=999) NSEGS,FR,TAU,PRINTW
WRITE(6,4) NSEGS
IF(NSEGS .LE. 0 .OR. NSEGS .GT. 15) GO TO 990
IF(FR .LE. 0. .OR. FR .GT. 1.) FR=.001
WRITE(6,5)FR
WRITE(6,150) TAU
M=0
V0=0.0

C
CCC READ AND WRITE INPUT PARAMETERS FOR EACH SEGMENT

C
DO 11 I=1,NSEGS
READ(5,12) NUMPTS(I),ITYP,ISIGN,ZEP,RHOEP,THETA
IF(ISIGN .EQ. BL) ISIGN=PLUS
WRITE(6,13) I,NUMPTS(I),ITYP,ISIGN,ZEP(1),RHOEP(1),ZEP(2),
& RHOEP(2)
IF(NUMPTS(I).LT.0 .OR. ITYP.LE.0 .OR. ITYP.GT.3) GO TO 990
IF(ITYP .NE. 3) WRITE(6,14) THETA
IF (NUMPTS(I).EQ.0) GOTO 11
N=M
THETA=PI*THETA/180.0
M=M+NUMPTS(I)
IF(M .GT. 102) GO TO 990
CALL DATA(ITYP,N,M,ZEP,RHOEP,THETA,FR,VOLINC)
IF(ISIGN .EQ. MIN) VOLINC=-VOLINC
V0=V0+VOLINC
11 CONTINUE

C
CCC SET UP LINEAR SYSTEMS OF EQUATIONS TO BE SOLVED

C
DO 2 N=1,M
INDX(1)=N
AN=ARC(N)
TN=RHO(N,8)

C
DO 3 L=N,M
IF(L .EQ. N) GO TO 82
AL=ARC(L)
TL=RHO(L,8)
AX11(N,L)=CMPLX(0.,0.)
AX11(L,N)=CMPLX(0.,0.)
AX33(N,L)=CMPLX(0.,0.)
AX33(L,N)=CMPLX(0.,0.)
INDX(2)=L

C
DO 103 J=1,3,2
JP6=J+6

C
DO 104 LL=1,2
I=3-LL
I1=INDX(LL)
I2=INDX(I)
TI2=RHO(I2,JP6)
CALL SETUP(I1,I2,JP6,AXI11,AXI33)
AX11(I1,I2)=AX11(I1,I2)-AXI11*TI2
AX33(I1,I2)=AX33(I1,I2)-AXI33*TI2

```
104          CONTINUE
C
103          CONTINUE
C
      CALL SETUP(N,L,8,AXI11,AXI33)
      AX11(N,L)=AL*(W1*AX11(N,L)-AXI11*W0*TL)
      AX33(N,L)=AL*(W1*AX33(N,L)-AXI33*W0*TL)
      CALL SETUP(L,N,8,AXI11,AXI33)
      AX11(L,N)=AN*(W1*AX11(L,N)-AXI11*W0*TN)
      AX33(L,N)=AN*(W1*AX33(L,N)-AXI33*W0*TN)
      GO TO 3
82          IF(FR .EQ. 1.0) GO TO 3
C
          DO 86 I=1,6
86          CALL SETUP(N,N,I,ST1(I),ST3(I))
C
          U=.5*(1.0-FR)*AN
          POM=PI*(CMPLX(1.,0.)+TAU)/(CMPLX(1.,0.)-TAU)
          AX11(N,N)=POM-U*(W0*(RHO(N,2)*ST1(2)+RHO(N,5)*ST1(5))
&          +W1*(RHO(N,1)*ST1(1)+RHO(N,3)*ST1(3)+RHO(N,4)*ST1(4)+
&          RHO(N,6)*ST1(6)))
          AX33(N,N)=POM-U*(W0*(RHO(N,2)*ST3(2)+RHO(N,5)*ST3(5))
&          +W1*(RHO(N,1)*ST3(1)+RHO(N,3)*ST3(3)+RHO(N,4)*ST3(4)+
&          RHO(N,6)*ST3(6)))
3          CONTINUE
C
2          CONTINUE
C
CCC        SOLVE MATRIX EQUATIONS AND INTEGRALS TO GET X11 AND X33
C
CCC        COMPUTE X11
C
          DO 20 I=1,M
20          B(I)=-TWOPI*RHO(I,8)
C
          CALL DECOMP(AX11,M)
          CALL SOLVE(AX11,X,B,M)
C
          IF (PRINTW.NE.1) GOTO 30
          WRITE(6,44) (I,Z(I,8),RHO(I,8),X(I), I=1,M)
30          CONTINUE
C
          DO 26 I=1,M
26          X(I)=RHO(I,8)*C(I,8)*X(I)
C
          CALL INTEG(X,NSEGS,NUMPTS,X11)
          X11=PI/V0*X11
C
CCC        COMPUTE X33
C
          DO 24 I=1,M
24          B(I)=-TWOPI*Z(I,8)
C
          CALL DECOMP(AX33,M)
          CALL SOLVE(AX33,X,B,M)
C
          IF (PRINTW.NE.1) GOTO 35
          WRITE(6,46) (I,Z(I,8),RHO(I,8),X(I), I=1,M)
35          CONTINUE
C
```

```
      DO 22 I=1,M
22     X(I)=-2.*RHO(I,8)*X(I)*S(I,8)
C
      CALL INTEG(X,NSEGS,NUMPTS,X33)
      X33=PI/V0*X33
C
CCC   WRITE OUTPUT
C
      WRITE(6,52)V0
      WRITE(6,200) X11
      WRITE(6,210) X33
      GO TO 37
990  WRITE(6,991)
999  STOP
C
CCC   FORMATS
C
34   FORMAT(I5,3E12.6,I5)
4    FORMAT('1***** PROGRAM DIELCOM *****',
&        '/0',5X,'SEGMENTS',4X,'=',I2)
1    FORMAT(' ',5X,'BODY #2',5X,'=',I2)
5    FORMAT(' ',5X,'EXCLUSION  =',F7.4)
150  FORMAT(' ',5X,'TAU          = ',E12.6,' +J',E12.6)
12   FORMAT(2I5,2X,A1,5E12.6)
13   FORMAT('0SEGMENT #',I2,';'/ ' ',5X,'CELLS',9X,'= ',I2/' ',5X,
& 'TYPE KEY',6X,'=',I2/' ',5X,'VOLUME SENSE = ',A4/' ',5X,
& '1ST END POINT = (',F10.5,', ',F10.5,')'/' ',5X,
& '2ND END POINT = (',F10.5,', ',F10.5,')')
14   FORMAT(' ',5X,'THETA (DEG)  =',F10.5)
52   FORMAT('0COMPUTED RESULTS;'/ ' ',5X,'VOLUME = ',F10.5)
200  FORMAT(' ',5X,'X11/V  = ',F10.5,' +J',F10.5)
210  FORMAT(' ',5X,'X33/V  = ',F10.5,' +J',F10.5)
44   FORMAT('0',6X,'I',11X,'Z',15X,'RHO',21X,'W1(I)'/
&        (' ',5X,I3,5X,2(F12.6,5X),E12.6,' +J',E12.6))
46   FORMAT('0',6X,'I',11X,'Z',15X,'RHO',21X,'W3(I)'/
&        (' ',5X,I3,5X,2(F12.6,5X),E12.6,' +J',E12.6))
991  FORMAT('0*** ERROR IN DATA')
      END
```

```
C
C*****C
C
C   SUBROUTINE DATA(IN,MX,MY,ZEP,RHOEP,THETA,B,VOL)
C
C*****C
C
C   SUBROUTINE DATA is called by the MAIN PROGRAM for each
C   segment profile. It computes the z and rho co-ordinates,
C   sin(alpha), cos(alpha), and the arc length for each cell.
C   Arguments:
C       IN      Type key for segment
C       MX      Total number of cells in segments to left
C       MY      MX + (number of cells in this segment)
C       ZEP     z - coordinate end points of segment:
C              ZEP(1)=z1, ZEP(2)=z2
C       RHOEP   Rho coordinate end points of segment:
C              RHOEP(1)=rho1, RHOEP(2)=rho2
C       THETA   Angle (in radians) subtended by circular
C              arc at its center
C       B       Fractional exclusion (beta)
C       VOL     Incremental volume of segment
C
C   DIMENSION ZEP(2),RHOEP(2)
C   COMMON RHO(102,9),Z(102,9),ARC(102),C(102,9),S(102,9)
C   DATA STEP/.3872988/
C
C   MXP1=MX+1
C   EN=FLOAT(MY-MX)
C   IF(B .NE. 1) SUBSTP=.5*(1.0-B)*STEP
C   IF(IN-2)1,2,3
C
C   CCC   TYPE KEY 'IN' = 1 (CONCAVE DOWN SEGMENT)
C
C   1 CC=-1.0
C   GO TO 10
C
C   CCC   TYPE KEY 'IN' = 2 (CONCAVE UP SEGMENT)
C
C   2 CC=1.0
C
C   CCC   COMPUTE THE RADIUS OF CURVATURE
C
C   10 ST2=SIN(THETA/2.0)
C   A=ZEP(2)-ZEP(1)
C   RAD=0.5*SQRT((RHOEP(1)-RHOEP(2))**2+A*A)/ST2
C
C   CCC   COMPUTE THE CENTER OF CURVATURE CO-ORDINATES
C
C   DD=A/ABS(A)
C   T=CC*DD*COS(THETA/2.)/ST2
C   ZCNT=0.5*(ZEP(1)+ZEP(2)+T*(RHOEP(1)-RHOEP(2)))
C   RHOCNT=0.5*(RHOEP(1)+RHOEP(2)+T*A)
C
C   CCC   COMPUTE THE INCREMENTAL VOLUME
C
C   U2=ZEP(2)-ZCNT
C   U1=ZEP(1)-ZCNT
C   VOL=3.141593*ABS(A*(RHOCNT**2+RAD*RAD-(U2**2+U1*U2+U1**2)/3.0))
```



```
& -CC*RHOCNT*(U2*(RHOEP(2)-RHOCNT)-U1*(RHOEP(1)-RHOCNT) +RAD*RAD
& *DD*THETA))
C
CCC COMPUTE COS(ALPHA), SIN(ALPHA), Z, AND RHO AT THE BOUNDARIES
CCC OF EACH CELL, AND THE ARC LENGTH OF EACH CELL IN THE SEGMENT
C
BETA=CC*DD*THETA/EN
THET1=ATAN2(RHOEP(1)-RHOCNT,ZEP(1)-ZCNT)
U=ABS(BETA*RAD)
B3=STEP*BETA
C
DO 902 I=MXP1,MY
PHI=THET1+(I-MX-.5)*BETA
IF(B .EQ. 1.0) GO TO 1905
C
DO 1902 J=1,2
ANG=PHI+.5*(J-1.5)*BETA*(1.0+B)
C
DO 1903 L=1,3
PSI=ANG+(L-2)*SUBSTP*BETA
M=L+3*(J-1)
C(I,M)=-CC*SIN(PSI)
S(I,M)=CC*COS(PSI)
Z(I,M)=ZCNT+RAD*CC*S(I,M)
1903 RHO(I,M)=RHOCNT-CC*RAD*C(I,M)
C
1902 CONTINUE
C
1905 DO 903 J=7,9
ANG=PHI+(J-8)*B3
C(I,J)=-CC*SIN(ANG)
S(I,J)=CC*COS(ANG)
Z(I,J)=ZCNT+RAD*CC*S(I,J)
903 RHO(I,J)=RHOCNT-CC*RAD*C(I,J)
C
902 ARC(I)=U
C
RETURN
C
CCC TYPE KEY 'IN' = 3 (LINEAR SEGMENT)
C
CCC COMPUTE COS(ALPHA), SIN(ALPHA), Z, AND RHO AT THE BOUNDARIES
CCC OF EACH CELL, AND THE ARC LENGTH OF EACH CELL IN THE SEGMENT
C
3 DX=(ZEP(2)-ZEP(1))/EN
DY=(RHOEP(2)-RHOEP(1))/EN
U =SQRT(DX*DX+DY*DY)
SI=DY/U
CI=DX/U
C
DO 917 I=MXP1,MY
PHI=FLOAT(I-MX)-.5
IF( B .EQ. 1.0) GO TO 1800
C
DO 1802 J=1,2
ANG=PHI+.5*(J-1.5)*(1.0+B)
C
DO 1803 L=1,3
M=L+3*(J-1)
PSI=ANG+(L-2)*SUBSTP
```

```
      Z(I,M)=ZEP(1)+PSI*DX
      RHO(I,M)=RHOEP(1)+PSI*DY
      S(I,M)=SI
1803      C(I,M)=CI
C
1802      CONTINUE
C
1800      DO 913 J=7,9
          ANG=PHI+(J-8)*STEP
          Z(I,J)=ZEP(1)+ANG*DX
          RHO(I,J)=RHOEP(1)+ANG*DY
          C(I,J)=CI
          S(I,J)=SI
    913
C
    917      ARC(I)=U
C
CCC      COMPUTE THE INCREMENTAL VOLUME
C
C
      VOL=1.047198*(ZEP(2)-ZEP(1))*(RHOEP(1)**2+RHOEP(1)*RHOEP(2)+
&      RHOEP(2)**2)
      RETURN
      END
```

```
C
C*****C
C
C      SUBROUTINE INTEG(V,NSEG,NUMPTS,SUM)
C
C*****C
C      SUBROUTINE INTEG is called by the MAIN PROGRAM to
C      numerically integrate quadratic interpolating polynomials
C      approximating the data on each segment of the profile.
C      Arguments:
C      V      Vector of function values, ordered as cells
C      NSEG   Total number of segments in the profile
C      NUMPTS Vector containing in NUMPTS(I) the number of
C            cells on the Ith segment: I = 1, NSEG
C      SUM    Integral of V across the profile
C
C      COMMON RHO(102,9),Z(102,9),ARC(102),C(102,9),S(102,9)
C      COMPLEX V(102),SUM
C      DIMENSION NUMPTS(15)
C      SUM=0.0
C      JACC=1
C
C      DO 3000 I=1,NSEG
C      T=ARC(JACC)
C      L=NUMPTS(I)
C      N=L+JACC-1
C      SUM=SUM+T*(0.625*(V(JACC)+V(N))-0.125*(V(JACC+1)+V(N-1)))
C      IF(L/2 .NE. (L+1)/2) GO TO 3001
C      SUM=SUM+T*(0.6666667*V(N-1)-0.08333333*V(N-2)+0.4166667*V(N))
3001  LM1=N-1
C      JLO=JACC+1
C
C      DO 3002 J=JLO,LM1,2
3002  SUM=SUM+0.3333333*T*(V(J-1)+4.0*V(J)+V(J+1))
C
3000  JACC=JACC+L
C
C      RETURN
C      END
```

```
C
C*****C
C
C      SUBROUTINE DECOMP(UL,N)
C
C*****C
C      SUBROUTINE DECOMP is called by the MAIN PROGRAM to perform
C      a L - U decomposition on the N x N matrix UL.
C
C      COMPLEX UL(102,102),PIVOT,EM,CDUM
C      COMMON /SOL/IPS(102)
C      CABS1(CDUM)=REAL(CDUM)+AIMAG(CDUM)
C
C      DO 5 I=1,N
5      IPS(I)=I
C
C      NM1=N-1
C
C      DO 16 K=1,NM1
C      BIF=0.0
C
C      DO 11 I=K,N
C      IP=IPS(I)
C      IF(CABS1(UL(IP,K)) .LE. BIF) GO TO 11
C      BIF=CABS1(UL(IP,K))
C      IDXPIV=I
11      CONTINUE
C
C      IF(IDXPIV .EQ. K) GO TO 15
C      J=IPS(K)
C      IPS(K)=IPS(IDXPIV)
C      IPS(IDXPIV)=J
15      KP=IPS(K)
C      PIVOT=UL(KP,K)
C      KP1=K+1
C
C      DO 16 I=KP1,N
C      IP=IPS(I)
C      EM=-UL(IP,K)/PIVOT
C      UL(IP,K)=-EM
C
C      DO 16 J=KP1,N
C      UL(IP,J)=UL(IP,J)+EM*UL(KP,J)
16      CONTINUE
C
C      RETURN
C      END
```

```
C
C*****C
C
C      SUBROUTINE SOLVE(UL,X,B,N)
C
C*****C
C
C      SUBROUTINE SOLVE is called by the MAIN PROGRAM to solve
C      the linear system  $UL * X = B$  by back substitution after
C      the UL matrix has been decomposed by the SUBROUTINE DECOMP.
C
C      COMPLEX UL(102,102),X(102),SUM
C      DIMENSION B(102)
C      COMMON /SOL/IPS(102)
C      NP1=N+1
C      IP=IPS(1)
C      X(1)=B(IP)
C
C      DO 2 I=2,N
C      IP=IPS(I)
C      IM1=I-1
C      SUM=CMPLX(0.,0.)
C
C      DO 1 J=1,IM1
1      SUM=SUM+UL(IP,J)*X(J)
C
C      2      X(I)=B(IP)-SUM
C
C      IP=IPS(N)
C      X(N)=X(N)/UL(IP,N)
C
C      DO 4 IBACK=2,N
C      I=NP1-IBACK
C      IP=IPS(I)
C      IP1=I+1
C      SUM=CMPLX(0.,0.)
C
C      DO 3 J=IP1,N
3      SUM=SUM+UL(IP,J)*X(J)
C
C      4      X(I)=(X(I)-SUM)/UL(IP,I)
C
C      RETURN
C      END
```

```
C
C*****C
C
C      SUBROUTINE ELLI (M1,K,E,KPR)
C
C*****C
C      SUBROUTINE ELLI is called by the SUBROUTINE SETUP to
C      compute the elliptic integrals K(m), E(m), and derivative
C      K'(m) from their power series approximations. (See
C      sec. 4.2 of "The Numerical Solution of Low Frequency
C      Scattering Problems".)
C
C      REAL M1,K,KPR
C      T=-ALOG(M1)
C      K=1.386294+.5*T+M1*(9.666344E-2+.1249859*T+M1*(3.590092E-2
C      & +6.880249E-2*T+M1*(3.742564E-2+3.328355E-2*T+M1*(1.451196E-2
C      & +4.41787E-3*T)))
C      E=1.0+M1*(.4432514+.2499837*T+M1*(6.260601E-2+9.20018E-2*T+M1*(
C      & 4.757348E-2+4.069698E-2*T+M1*(1.736506E-2+5.264496E-3*T)))
C      KPR=.5/M1 + 2.83225E-2 - .1249859*T + M1*(-2.999362E-3-.137605*T
C      & +M1*(-7.899336E-2 - 9.985066E-2*T + M1*(-5.362998E-2 -
C      & 1.767148E-2 * T )))
C      RETURN
C      END
```

```

C
C*****
C
SUBROUTINE SETUP(I,J,L,AXI11,AXI33)
C
C*****
C
SUBROUTINE SETUP is called by the MAIN PROGRAM to compute
C the quantities AXI11 (f(i,j) of eq. 227 in "The Numerical
C Solution of Low Frequency Scattering Problems") and AXI33
C (a similar quantity for x33).
C Arguments:
C I Subscript of observer (unprimed) cell
C J Subscript of remote (primed) cell
C L Index of the point within remote cell for
C which the kernels are to be computed
C (See SUBROUTINE DATA)
C AXI11 See above
C AXI33 See above
C
COMMON RHO(102,9),Z(102,9),ARC(102),C(102,9),S(102,9)
REAL M,M1,K,KPR
REAL*8 DA1,DA2,DM1,DZD,DRP,DR,DRRP
DZD=DBLE(Z(J,L))-DBLE(Z(I,8))
ZD=SNGL(DZD)
R=RHO(I,8)
DR=DBLE(R)
RP=RHO(J,L)
DRP=DBLE(RP)
DRRP=DR*DRP
RRP=SNGL(DRRP)
DA1=DRRP+DRRP
DA2=DR*DR+DRP*DRP+DZD*DZD
DM1=(DA2-DA1)/(DA2+DA1)
M1=SNGL(DM1)
IF (ABS(M1).LT.1.E-50) WRITE(6,100) M1,Z(J,L),Z(I,8),R,RP,DA1,DA2
100 FORMAT(7E10.3)
M=1.-M1
CALL ELLI(M1,K,E,KPR)
A0=M/RRP
A1=SQRT(A0)
A2=M+M
A3=2.-M
A1=A1*A0
A3=.5*A3
A0=C(J,L)
OM0=.25*A1*(K+2.*M*KPR)
OM1=-A1*(.25*K-A3*KPR)
OM2=A1*(E-A3*((A3+M)*K - A2*A3*KPR))/(M*M)
AXI11=R*A0*OM2+(ZD*S(J,L)-RP*A0)*OM1
AXI33=R*A0*OM1+(ZD*S(J,L)-RP*A0)*OM0
RETURN
END

```

C
CCC MODIFICATIONS OF MINK TO DIELCOM BY TOM WILLIS
C

References

- [1] T.B.A. Senior, (1976), Low frequency scattering by a dielectric body, Radio Sci., Vol. 11 (5), pp. 447-482.
- [2] T.B.A. Senior and D. J. Ahlgren (1972), The Numerical Solution of Low Frequency Scattering Problems, The University of Michigan Radiation Laboratory Report No. 013630-9-T.
- [3] G. Forsythe and C. B. Moler, (1967), Computer Solution of Linear Algebraic Systems, Prentice Hall, Englewood Cliffs; pp. 68-69.
- [4] M. Abramowitz and I. A. Stegun (eds.), (1964), Handbook of Mathematical Functions, NBS Appl. Math. Series No. 55; p. 590.